
1.   Introduction





From our knowledge, little or any software programs are actually available to simulate adaptive testing. One of these is POSTSIM distributed by Assessment Systems Corporation. POSTSIM is mostly limited by the incapacity to allow the user to modify the program or to add new routines. In fact, scientists exploring the characteristics of adaptive tests usually elaborate these programs themselves, and for themselves. Consequently, the access to these programs by the research community is difficult. Also, when available, they are not versatile, nor is the source code that goes with them in order to favour adaptations.





To palliate this situation and support adaptive testing research, such a program is proposed here, with the SAS language related source code. A preliminary version of this program was used before by Raîche (2000) to study the sampling distribution of the proficiency level in adaptive testing according to two stopping rules: having regard to the number of administered items and to the standard error of the proficiency level estimate. The new version gives access to improvements as regards to the program versatility and to new proficiency level estimation methods. The Rasch one parameter modelization is the one retained. It would be fairly easy to accommodate the program for other dichotomous response modelizations, like the two, three and four parameters logistic or normal models. 





Expected a posteriori proficiency level estimation method (EAP) is applied to compute provisory and final proficiency level estimates. The new proficiency level estimation methods are all adaptive modifications brought to the EAP method (Blais and Raîche, 2002, 2005; Raîche, 2004; Raîche and Blais, 2001, 2002, 2005). These methods are all adaptive in the a priori proficiency level estimation, the proficiency level estimation bias correction, the integration interval, or a combination of these factors. The use of these adaptive EAP estimation methods diminish considerably the shrinking, and so biasing, effect of the a priori proficiency level estimate encountered when this one is fixed at a constant value independantly of the previous provisory value computed of the proficiency level.





Another of the program’s peculiarity is the feasibility to compare theoretical values of the standard error, skewness, and kurtosis of the estimated proficiency level with the empirical values of these statistics. This, according to predetermined sampling distributions of the proficiency level estimate. 








2.    Program Description





2.1  Variables Initialisation 





In the first section of the program, some variables have to be specifically initialized from the beginning. So, the maximum number of administered items (item) and the subject number (sujet) have to be fixed in advance by the user. If the maximum number of administered items is over 60, it is recommended to increase the quadrature point number (qpoint), since in this situation skewness and kurtosis coefficients need a greater number of quadrature points than it is required with usual estimations in adaptive testing. Raîche (2000, p. 136-147) gives explanations and indications about the quadrature point number required.





In most cases, other values do not need to be modified. For example, Haley’s constant (D) can stay fixed at 1.702 to warrant that cumulative probabilities obtained by the logistic probability function do not differ more than 0.01 with the cumulative probabilities obtained by the normal probability function N(0,1). Nevertheless, one can modify the integration interval (thetamin, thetamax), the a priori proficiency level estimation (baye), or the values of the generating proficiency level (minhab, maxhab, byhab). 





However, one peculiarity of the program is to offer the opportunity to study the adaptive EAP estimation methods proposed by Raîche and Blais (2002, 2005): adaptive a priori proficiency level estimate (aap), adaptive bias correction of Bock et Mislevy (1982) (acb), or adaptive integration interval (aii). Bock and Mislevy’s correction can also, as usual, be applied only at the end of the test administration (fcb). When the global variables aap, acb, aii or fcb are fixed at 0, the adaptive test develops as usual, without applying these adaptive procedures. To use them, one has to fix at 1 each of the associated global variables. 








2.2  Fonction Description (Macros)





The %Prior macro computes the probability that a subject has a specific proficiency level when the mean of the a priori distribution is fixed at the a priori proficiency level estimate (baye) and that the standard error of the proficiency level is equal to 1: a normal distribution N(baye, 1) is used. 





The %Logist macro computes the probability of the response to each item (0 for an incorrect response, 1 for a correct response) according to the Rasch one parameter modelization. It is the %Reponse macro that generates a random correct or incorrect response to the item according to the provisory proficiency level estimate (Harwell, Stone, Hsu and Kirisci, 1996, p. 116).





To settle the integration step from the quadrature point number, qpoint, and from the integration interval (thetamin - thetamax), the %Step macro is called by the %Mu2 macro. From this last macro, are computed the first central moment, i.e. the proficiency level estimate, like the second, third, and fourth central moments, i.e the standard error, the skewness, and the kurtosis of the sampling distribution of the proficiency level estimate. Equations are described by Raîche (2000), and Blais and Raîche (2002, 2005). 





Finally, the %Sitem and %Sseo macros start the analysis of the simulation results according to the the stopping rule at study: having regard to the administered number of items (%Sitem) or having regard to the standard error of the proficiency level estimate (%Sseo).

















3.   Output Examples





As an example, presented in table 2, are the results obtained by a simulation of 500 adaptive test administrations. The generating proficiency level is fixed at an extreme value of –3, and the standard error retained as a stopping rule varies between 0.50 and 0.20 by steps of 0.05. Results obtained when the stopping rule according to the administered items number (from 1 to 40 by steps of 1) are also presented in table 3. Beforehand, in table 1, output variables are described.











Table 1. Output Variables computed from the work.thesd_&nitem data set


�
�



RSEO


ITEM


MTEO


SE


MSEO


AS


MASO


KU


MKUO�



Standard error retained for the stopping rule (when %RSEO is used)


Administered items number retained for the stopping rule (when %RITEM is used)


Mean of the proficiency level obtained for %SUJET subjects


Empirical standard error computed from the estimated values of the proficiency level


Mean of the theoretical standard error of the proficiency level estimate 


Empirical skewness computed from the estimated values of the proficiency level


Mean of the theoretical skewness of the proficiency level estimate


Empirical kurtosis computed from the estimated values of the proficiency level


Mean of the theoretical kurtosis of the proficiency level estimate


�
�









�






Table 2. Adaptive testing simulation according to the standard error of the proficiency level estimate stopping rule 








--------------------------------------- PROFICIENCY = -3 ------------------





  OBS RSEO  SE       MSEO    MTEO     AS      MASO      KU      MKUO





    1 20    0.19574  0.19843 -2.86735 0.11396 -0.04698  0.25201 0.049934


    2 25    0.25252  0.24701 -2.80753 0.27885 -0.07072  0.17099 0.066383


    3 30    0.31028  0.29521 -2.73056 0.30850 -0.09111 -0.04540 0.067628


    4 35    0.36666  0.34223 -2.63158 0.40457 -0.10454  0.17772 0.058355


    5 40    0.41216  0.38983 -2.50978 0.31117 -0.11697 -0.04393 0.048892


    6 45    0.41746  0.43603 -2.36559 0.59112 -0.14301  0.62322 0.059110


    7 50    0.47513  0.48159 -2.18776 0.22684 -0.15117 -0.83537 0.047787





�
�















Table 3. Adaptive testing simulation according to the administered items number stopping rule








--------------------------------------- PROFICIENCY = -3 ---------------





  OBS  ITEM MTEO     SE      MSEO    AS      MASO     KU      MKUO





    1   1   -0.55234 0.12336 0.82605 8.99056 -0.10337 79.1467 0.12845


    2   2   -0.96589 0.14864 0.72751 6.75061 -0.18018 44.6226 0.16464


    3   3   -1.29344 0.18587 0.66402 4.65296 -0.21728 21.1136 0.16994


    4   4   -1.55309 0.23565 0.61625 2.89965 -0.22737  7.6644 0.15817


    5   5   -1.77271 0.27796 0.57780 2.02508 -0.22464  3.1483 0.13351


    6   6   -1.94580 0.31936 0.54350 1.36880 -0.21067  0.8764 0.10341


    7   7   -2.09104 0.34974 0.51299 0.93989 -0.19322 -0.0559 0.07467


    8   8   -2.21253 0.37128 0.48554 0.71883 -0.17540 -0.0066 0.05362


    9   9   -2.31790 0.37892 0.46109 0.52809 -0.16050 -0.0362 0.04202


   10  10   -2.39150 0.37568 0.43810 0.37986 -0.14888 -0.0532 0.04436


   11  11   -2.44573 0.37099 0.41748 0.37734 -0.14064  0.0139 0.04928


   12  12   -2.51386 0.36117 0.40056 0.31944 -0.13420  0.1032 0.04622


   13  13   -2.55526 0.35980 0.38407 0.27246 -0.12506  0.0225 0.04579


   14  14   -2.59567 0.35359 0.36956 0.32091 -0.12045 -0.0922 0.04448


   15  15   -2.63012 0.34918 0.35602 0.31528 -0.11480 -0.0375 0.04594


   16  16   -2.65362 0.33411 0.34373 0.34075 -0.11288  0.0050 0.05310


   17  17   -2.67329 0.33009 0.33202 0.27598 -0.10729 -0.1235 0.05489


   18  18   -2.68974 0.32909 0.32113 0.24310 -0.10172 -0.1183 0.05702


   19  19   -2.70290 0.31428 0.31132 0.31374 -0.10050  0.0152 0.06323


   20  20   -2.72380 0.30870 0.30270 0.33443 -0.09818  0.0436 0.06402


�
�

SIMTAO 1.1 - Source Code (SAS 6.0)





/***************************************************************************/;


/***************************************************************************/;


/**                                                                       **/;


/**                      Adaptive Testing Simulation                      **/;


/**                                                                       **/;


/**                              SIMTAO 1.1                               **/;


/**                                                                       **/;


/**                        Coded by Gilles Raîche                         **/;


/**                         Raiche.Gilles@uqam.ca                         **/;


/**                                                                       **/;


/**                    Université du Québec à Montréal                    **/;


/**                  Faculté des sciences de l’éducation                  **/;


/**                  Département d’éducation et pédagogie                 **/;


/**                                                                       **/;


/**                        Montréal, QUÉBEC, CANADA                       **/;


/**                                                                       **/;


/**                             April 2005                                **/;


/**                                                                       **/;


/***************************************************************************/;


/***************************************************************************/;





 Options replace nodate pageno=1 pagesize=256 nosource nonotes


         nomtrace nomlogic nomprint;





/* Constant Initialisation *************************************************/;


 %Let nitem=10   ;/* Item Number                                           */;


 %Let qpoint=40  ;/* Quadrature Points                                     */;


 %Let D=1.702    ;/* Haley’s Constant                                      */;


 %Let thetamin=-4;/* Inferior EAP Limit of Integration                     */;


 %Let thetamax= 4;/* Superior EAP Limit of Integration                     */;


 %Let bayes=0    ;/* A Priori Proficiency Level                            */;


 %Let sujet=100  ;/* Subject number                                        */;


 %Let minhab= -3 ;/* Minimum Proficiency level                             */;


 %Let maxhab=  3 ;/* Maximum Proficiency level                             */;


 %Let byhab=   1 ;/* By Steps of ...                                       */;


 %Let aap=0      ;/* A Priori Adaptive Estimation Method (NO=0;YES=1)      */;


 %Let acb=0      ;/* Adaptive Bock and Milslevy Bias Correction(NO=0;YES=1)*/;


 %Let fcb=0      ;/* Final Bock and Milslevy Bias Correction (NO=0;YES=1)  */;


 %Let ain=0      ;/* Adaptive integration Interval (NO=0;YES=1)            */;


/* *************************************************************************/;





 TITLE1 "ADAPTIVE TEST SIMULATION (n=&SUJET)";


 TITLE2 "(ADAPTIVE ESTIMATION : AAP=&AAP, AIN=&AIN, ACB=&ACB, FCB=&FCB)";





 %Macro Prior (Theta,mu);


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* A priori Probability Function N(mu,1) used by Bock and Mislevy (1982)   *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: theta, mu                                             *;


 %* Global Variables: step                                                  *;


 %* Function Call   : none                                                  *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


  Probnorm(&Theta-&mu + step/2) - Probnorm(&Theta-&mu - step/2);;


 %Mend;





 %Macro logist (Theta,b,D,r);


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* One Parameter Logistic Function                                         *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: theta, b, r                                           *;


 %* Global Variables: D                                                     *;


 %* Function Call   : none                                                  *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


  ((((1+exp(-&D*(&Theta-&b)))**-1))**&r) *


  (1-(((1+exp(-&D*(&Theta-&b)))**-1)))**(1-&r);


 %Mend;





 %Macro Reponse (i,habilete);


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Random Item Response according to the Generating Proficiency Level      *;


 %* Method Used by Harwell, Stone, Hsu and Kirisci (1996)                   *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: i, habilete                                           *;


 %* Global Variables: D, b                                                  *;


 %* Function Call   : none                                                  *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


  plog= exp(&D*(&habilete-b{&i})) / (1+exp(&D*(&habilete-b{&i})));


  puni=ranuni(1);


  r{&i}=plog > puni;


 %Mend;





 %Macro step;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Class Width According to Quadrature Point Number and to a priori        *;


 %* Function Sum Surface on the interval Thetamin … Thetamax                *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: surface, j                                            *;


 %* Global Variables: step, thetamax, thetamin, qpoint                      *;


 %* Function Call   : prior                                                 *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


  surface=0;


  Do j=1 to &nitem;


     %* Class Width According to Quadrature Point Number   ..................;


     step=(&thetamax-&thetamin)/(&qpoint-1);


     end;


  %* A priori Function Sum Surface  .........................................;


  Do theta=&thetamin to &thetamax by step;


     surface=surface + %Prior(theta,0);


     end;


  %* Put "surface = " surface; *;


 %Mend;





 %Macro MU2(habilete);


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Moment Estimation by EAP method                                         *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: b1,j,num,p,lnp,i,logist,prior                         *;


 %* Global Variables: nitem,habilete,thetamin,thetamax,step,theta,b,D,r     *;


 %*                      bayes,te,denom                                     *;


 %* Function Call   : reponse,logist,prior                                 *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


  %* Difficulty Parameter of the First Item  ................................;


  b1=0;


  %* Numerators and Denominators Initialisation to 0,    ....................;


  %* Like the A Priori Proficiency Level Estimate ...........................;


  Do j=1 to &nitem;


     num(j)=0;denom(j)=0;bayes(j)=&bayes;


     num2_(j)=0;num3_(j)=0;num4_(j)=0;


     end;


  %* j Proficiency Level Estimation      ....................................;


  Do j=1 to &nitem;                                       %* Begin Level 1  .;


     %* Adaptive Parameters for Numeric Integration        ..................;


     ain=0;If &ain=1 then do; ain=b(j); end;


     %Reponse(j,&habilete);


     %* A Posteriori Proficiency level Expectation              .............;


     Do theta=(&thetamin+ain) to (&thetamax+ain) by step; %* Begin Level 2  .;


        p=1;lnp=log(p);


        Do i=1 to j;                                      %* Begin Level 3  .;


           logist=%logist(theta,b(i),&D,r(i));


           lnp=lnp+log(logist);


           end;                                           %* End Level 3    .;


        if j>1 then do; prior=%prior(theta,bayes(j-1));end;


        if j=1 then do; prior=%prior(theta,&bayes);end;


        num(j)=num(j)+     (theta)*exp(lnp+log(prior));


        denom(j)=denom(j)+  exp(lnp+log(prior));


        end;                                              %* End Level 2    .;


     %* jth Proficiency level Estimation      ...............................;


     te(j)=num(j)/denom(j);








     Do theta=(&thetamin+ain) to (&thetamax+ain) by step; %* Begin Level 2  .;


        p=1;lnp=log(p);


        Do i=1 to j;                                      %* Begin Level 3  .;


           logist=%logist(theta,b(i),&D,r(i));


           lnp=lnp+log(logist);


           end;                                           %* End Level 3    .;


        if j>1 then do; prior=%prior(theta,bayes(j-1));end;


        if j=1 then do; prior=%prior(theta,&bayes);end;


        num2_(j)=num2_(j)+((theta-te(j))**2)  *exp(lnp+log(prior));


        num3_(j)=num3_(j)+((theta-te(j))**3)  *exp(lnp+log(prior));


        num4_(j)=num4_(j)+((theta-te(j))**4)  *exp(lnp+log(prior));


        end;                                              %* End Level 2    .;


     %* jth Standard Error Estimation of proficiency level     ..............;


     se(j)=(num2_(j)/denom(j))**.5;


     %* jth Skewness Coefficient Estimation .................................;


     as(j)=(num3_(j)/denom(j))/se(j)**3;


     %* jth Kurtosis Coefficient Estimation .................................;


     ku(j)=((num4_(j)/denom(j))/se(j)**4)-3;


     denom(j)=denom(j)/surface;


     %* Adaptive Correction of the A Posteriori proficiency level ...........;


     if &acb=1 then do; te(j)=te(j)/(1-(se(j)*se(j))); end;


     %* Adaptative Estimation of the A Posteriori proficiency level .........;


     if &aap=1 then do; bayes(j) = te(j);end;


     if j<&nitem then do;


        b(j+1) =te(j);


        end;





     end;                                                 %* End Level 1    .;


 %Mend;














 %MACRO SITEM;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Theoretical and empirical values of the Estimators According to         *;


 %* Different Values of the Stopping Rule According to the Number of Item   *;


 %* Administered                                                            *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: none                                                  *;


 %* Global Variables: sujet, nitem                                          *;


 %* Function Call   : none                                                  *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 Proc means data=thesd_&nitem noprint;


      class habilete item;


      var seo aso kuo;


      output out=res1 mean(seo)=mseo mean(aso)=maso mean(kuo)=mkuo;


      run;


 Proc sort;by habilete item;run;





 Proc means data=thesd_&nitem noprint;


      class habilete item;


      var teo;


      output out=res2 mean=mteo std=se skewness=as kurtosis=ku;


      run;


 Proc sort;by habilete item;run;





 Data results;


      merge res1 res2;


      by habilete item;


      if (habilete NE "" and item NE "");


      run;


 Proc sort;by habilete;run;





 Proc print;


      by habilete;


      var item mteo se mseo as maso ku mkuo;


      run;


 %MEND;








 %Macro SSEO;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Theoretical and empirical values of the Estimators According to         *;


 %* Different Values of the Stopping Rule According to the                  *;


 %* Standard Error of the Proficiency level                                 *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 %* Local  Variables: h, i, j                                               *;


 %* Global Variables: sujet, nitem, minhab, maxhab, byhab                   *;


 %* Function Call   : none                                                  *;


 %* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


  %DO H=50 %TO 20 %BY -5;


    %DO I=&minhab %TO &maxhab %BY &byhab;


      %DO J=1 %TO &SUJET;


        Data seo40; set thesd_&nitem ;


             If (habilete=&I and sujet=&J and seo<=(&H/100)) then do;


                rseo=&H;


                output seo40;


                stop;


                end;


                Run;


        Proc datasets nolist; append base=sorti data=seo40;run;


      %END;


    %END;


  %END;





  Proc means data=sorti noprint;


       class habilete rseo;


       var  seo aso kuo;


       output out=res1 mean(seo)=mseo mean(aso)=maso mean(kuo)=mkuo;


       run;


  Proc sort;by habilete rseo;run;





  Proc means data=sorti noprint;


       class habilete rseo;


       var teo;


       output out=res2 mean=mteo std=se skewness=as kurtosis=ku;


       run;


  Proc sort;by habilete rseo;run;





  Data results;


       merge res1 res2;


       by habilete rseo;


       if (habilete NE "" and rseo NE "");


       run;


  Proc sort;by habilete;run;





  Proc print;


       by habilete;


       var rseo se mseo mteo as maso ku mkuo;


       run;


  Proc datasets;delete sorti;run;


 %MEND;





 /* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 /* DATA SET STORING INDIVIDUAL RESULTS                                     *;


 /* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


Data work.thesd_&nitem;


      array te[&nitem];     /* Proficiency Level Estimate .................*/;


      array r[&nitem];      /* Response Vector     ........................*/;


      array b[&nitem];      /* Difficulty Parameters    ...................*/;


      array num[&nitem];    /* Numerator for EAP (MU)   ...................*/;


      array num2_[&nitem];  /* Numerator for EAP (SE)   ...................*/;


      array num3_[&nitem];  /* Numerator for EAP (MC3)   ..................*/;


      array num4_[&nitem];  /* Numerator for EAP (MC4)   ..................*/;


      array denom[&nitem];  /* Denominator for EAP   ......................*/;


      array se[&nitem];     /* Standard Error .............................*/;


      array as[&nitem];     /* Skewness  ..................................*/;


      array ku[&nitem];     /* Kurtosis ...................................*/;


      array prc[&nitem];    /* Correct Responses Proportion  ..............*/;


      array bayes[&nitem];  /* A Priori Estimator  ........................*/;


      length default=3      /* Variable Format Storage         ............*/;


      b1=0;                 /* Difficulty Level Initialisation of b1   ....*/;


      %STEP;


      Do sujet = 1 to &sujet;                             %* Begin Level 1  .;


         Do habilete = &minhab to &maxhab by &byhab;      %* Begin Level 2  .;


         %MU2(habilete);


         /* Different Statistics According to the Number of ...............*/;


         /* Administered Items  ...........................................*/;


         Do item=1 to &nitem;                             %* Begin Level 3  .;


    


           ro=r{item};bo=b{item};teo=te{item};prco=prc{item};baye=bayes{item};


           seo=se{item};aso=as{item};kuo=ku{item};


           if &fcb=1 then do;teo=teo/(1-seo*seo);end;


           biais = teo-habilete;


           Output;


           end;                                           %* End Level 3  ...;


         End;                                             %* End Level 2  ...;


         End;                                             %* End Level 1  ...;


      Keep sujet item ro bo habilete teo prco seo aso kuo biais baye;


      run;








 /* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 /* TO PRINT INDIVIDUAL RESULTS,                                            *;


 /* TAKE OUT COMMENT CODE                                                   *;


 /* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *;


 /* Proc print data = work.thesd_&nitem;                                    *;








 %SSEO;


 %SITEM;

